/*
 *  Licensed to the Apache Software Foundation (ASF) under one or more
 *  contributor license agreements.  See the NOTICE file distributed with
 *  this work for additional information regarding copyright ownership.
 *  The ASF licenses this file to You under the Apache License, Version 2.0
 *  (the "License"); you may not use this file except in compliance with
 *  the License.  You may obtain a copy of the License at
 *
 *      https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing, software
 *  distributed under the License is distributed on an "AS IS" BASIS,
 *  WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 *  See the License for the specific language governing permissions and
 *  limitations under the License.
 */

import java.time.Instant
import java.time.LocalDate
import java.time.ZoneOffset
import java.time.format.DateTimeFormatter

Properties props = new Properties()
file('../gradle.properties').withInputStream {
    props.load(it)
}

allprojects {
    props.forEach { k, v -> project.ext.set(k as String, v) }

    apply from: rootProject.layout.projectDirectory.file('../dependencies.gradle')
}

ext {
    buildInstant = java.util.Optional.ofNullable(System.getenv("SOURCE_DATE_EPOCH"))
            .filter(s -> !s.isEmpty())
            .map(Long::parseLong)
            .map(Instant::ofEpochSecond)
            .orElseGet(Instant::now)
    formattedBuildDate = System.getenv("SOURCE_DATE_EPOCH") ?
            DateTimeFormatter.ISO_INSTANT.format(buildInstant) :
            DateTimeFormatter.ISO_DATE.format(LocalDate.ofInstant(buildInstant as Instant, ZoneOffset.UTC))
    buildDate = (buildInstant as Instant).atZone(ZoneOffset.UTC) // for reproducible builds

    isReleaseVersion = Boolean.parseBoolean(System.getenv('GRAILS_PUBLISH_RELEASE'))
    isCiBuild = System.getenv().containsKey('CI')
    configuredTestParallel = findProperty('maxTestParallel') as Integer ?: (isCiBuild ? 3 : Runtime.runtime.availableProcessors() * 3/4 as int ?: 1)
}

apply {
    from layout.projectDirectory.file('gradle/publish-root-config.gradle')
}
